

## Utilitaire pour Super-Oric "RCLORES0.COM" en Basic 1.1 Redéfinition des caractères en LORES 0

par André

En matière de couleurs, les possibilités du Super-Oric sont assez géniales. Par exemple, il est possible de redéfinir les caractères (8x8 pixels) soit en 4 couleurs (LORES 0) soit en 16 couleurs (LORES 1). La couleur de chaque pixel peut être redéfinie individuellement, sans effets de proximité et les 4 ou 16 couleurs utilisées peuvent être choisies parmi des milliers (32768 possibilités plus exactement). Nous avons déjà effectué quelques essais de redéfinitions des couleurs (voir les mag n°179 pages 13-15 et 180 pages 9-12), de redéfinition de caractères en LORES 0 (voir les mag 181 pages 9-12 et 182 pages 8-10) et en LORES 1 (voir le mag 183 pages 9-14).

Mais si la redéfinition des caractères de l'Oric (6x8 pixels en 2 couleurs) était relativement facile, celle des caractères du Super-Oric est plus ardue en LORES 0 (8x8 pixels en 4 couleurs) et surtout en LORES 1 (8x8 pixels en 16 couleurs). J'ai déjà indiqué des recettes pour effectuer la chose manuellement, mais c'est quand même un peu galère. Je vous propose donc aujourd'hui un petit utilitaire qui effectuera automatiquement le calcul des

DATA à incorporer dans la commande DEFCHAR pour redéfinir des caractères en LORES 0. Dans un prochain article viendra l'équivalent pour LORES 1.

**Cet utilitaire, qui est écrit en basic 1.1, tourne aussi bien sur Atmos que sur Super-Oric.** Mais évidemment, dans ce dernier cas, il faut disposer d'un clavier pour taper les valeurs de couleurs des pixels.

Je fais pour l'instant l'impasse sur la partie artistique de l'exercice, à savoir le dessin du ou des caractères à redéfinir. Il vous faudra colorier manuellement une grille de 8x8 cases en utilisant 4 crayons de couleurs. Ces couleurs se voient attribuer un n° de 0 à 3 et on obtient alors très facilement une grille de 8x8 cases, chacune étant numérotée de 0 à 3. L'utilitaire RCLORES0 dont vous trouverez le listing ci-dessous vous demande la liste de ces numéros, ligne par ligne et vous affiche la ligne DEFCHAR à incorporer dans votre programme pour Super-Oric. Notez qu'il se produit une courte pose à la fin de chaque ligne car le programme effectue les calculs ligne par ligne.

```

1000 ' *** REDEFCHAR ***
1010 ' (c) Andr{
1020 ' V1811061832
1030 ' SAVEU"RCLORES0"
1040 DIM L(7,7):'Tableau des couleurs des 8 pixels des 8 lignes
1050 DIM C(7,7,1):'Plans 0 et 1 des couleurs des 8x8 pixels du Caract}re
1060 DIM W(8):' Liste des Words pour la ligne DEFCHAR
1070 ' *** Titre
1080 CLS:PRINT"REDEFCHAR":PRINT"pour Super-Oric"
1090 PRINT"(C) Andr{ Ch{ramy 2006":PRINT
1100 PRINT"Code ASCII du caract}re"
1110 PRINT"(de 0 @ 1023)";:INPUT CA
1120 PRINT"Pour chacune des 8 lignes,"
1130 PRINT"entrez la couleur des 8 pixels":PRINT"de red{finition"
1140 PRINT"Valeurs de 0 @ 3 en LORES0"
1150 ' *** D{but de boucle pour 8 lignes
1160 FOR LIGNE=0 TO 7
1170 ' *** Remplissage des couleurs des pixels de la ligne
1180 PRINT:PRINT"Couleurs pour la ligne N";LIGNE
1190 FOR P=0 TO 7:' 8 Pixels
1200 PRINT"Pixel"P" ";:INPUT L(P,LIGNE)
1210 NEXT P
1220 ' *** Chaque bit est plac{ dans un plan distinct
1230 ' *** et se voit attribu{ une valeur V d{pendant de sa position

```

```

REDEFCHAR                                CAPS
pour Super-Oric
(C) André Chéramy 2006

Code ASCII du caractère
(de 0 à 1023)? 65
Pour chacune des 8 lignes,
entrez la couleur des 8 pixels
de redéfinition.
Valeurs de 0 à 3 en LORES0

Couleurs pour la ligne N 0
Pixel 0 ? 0
Pixel 1 ? 1
Pixel 2 ? 2
Pixel 3 ? 0
Pixel 4 ? 0
Pixel 5 ? 1
Pixel 6 ? 2
Pixel 7 ? 3

Couleurs pour la ligne N 1
Pixel 0 ? █

```

```

Pixel 6 ? 3
Pixel 7 ? 0

Couleurs pour la ligne N 6
Pixel 0 ? 2
Pixel 1 ? 3
Pixel 2 ? 0
Pixel 3 ? 1
Pixel 4 ? 2
Pixel 5 ? 0
Pixel 6 ? 1
Pixel 7 ? 1

Couleurs pour la ligne N 7
Pixel 0 ? 3
Pixel 1 ? 0
Pixel 2 ? 1
Pixel 3 ? 2
Pixel 4 ? 3
Pixel 5 ? 0
Pixel 6 ? 1
Pixel 7 ? 2

DEFCHAR#41,#3355,#66AA,#CC55,#99AA,#33
55,#66AA,#CC55,#99AA
Ready
█

```

```

1240 ' *** dans la colonne de d{finition du caract}re
1250 V=8:' Initialisation pour colonne 0
1260 M=1:' Masque pour b0
1270 FOR P=0 TO 7:' Pixels de 0 @ 7
1280 CP=L(P,LIGNE):'Lecture Couleur Pixel
1290 FOR B=0 TO 1:' bits de 0 @ 1
1300 IF(CP AND M) > 0 THEN C(0,P,B)=V ELSE C(0,P,B)=0
1310 M=M*2
1320 NEXT B
1330 M=1:' Reinitialisation lorsque tous les bits ont {t{ faits
1340 V=V/2:IF V<1 THEN V=8:' V prend les valeurs 8 4 2 1 8 4 2 1
1350 NEXT P
1360 ' *** Calcul du Word caract{risant la ligne en cours
1370 C(0,7,0)=C(0,7,0)+C(0,6,0)+C(0,5,0)+C(0,4,0)
1380 C(0,3,0)=C(0,3,0)+C(0,2,0)+C(0,1,0)+C(0,0,0)
1390 C(0,7,1)=C(0,7,1)+C(0,6,1)+C(0,5,1)+C(0,4,1)
1400 C(0,3,1)=C(0,3,1)+C(0,2,1)+C(0,1,1)+C(0,0,1)
1450 W(LIGNE)=C(0,7,0)+16*C(0,3,0)+256*C(0,7,1)+4096*C(0,3,1)
1470 NEXT LIGNE
1480 PRINT:PRINT"DEFCHAR"HEX$(CA) ;
1490 FOR LIGNE=0 TO 7
1500 PRINT" , "HEX$(W(LIGNE)) ;
1510 NEXT LIGNE
1520 END

```

